IV.5.3 Todistussemantiikka

Todistusmenetelmän pitää olla oikeellinen: sillä voi todistaa vain varmoja asioita.

Varmat asiat ovat ne, jotka ovat tosia kaikissa malleissa.

Kalvojen IV.5.2 mukaan Prolog-ohjelmalla L on pienin malli ML joka on samalla sen merkitys.

Kysymykseen ”Onko maalifakta p ∈ ML?” – eli kalvojen IV.5.1 aliohjelmakutsuun p – voi siis yrittää vastata todistamalla p väitteistä L.

Valitaan täydellinen todistussääntö eli sellainen jolla pystyy todistamaan kaiken sen, mikä on varmaa.

Kalvojen IV.5.2 todistuksen etsinnän epätäydellisyyteen se ei auta: yleisessä tapauksessa on mahdotonta tunnistaa millään algoritmilla, että todistusta ei ole.

Prolog-ohjelmien todistussäännön perusaskeleena käytetään resoluutiota. Sen muuttujaton muoto on

looginen ehto 1: q1 ∧ . . . ∧ qm → p1 ∨ . . . ∨ pn

looginen ehto 2: q′1 ∧ . . . ∧ q′m_′ → p′1 ∨ . . . ∨ p′n_′

sivuehto: p1 ja q′1 ovat samat

johtopäätös: q1 ∧. . .∧qm ∧q′2 ∧. . .∧q′m_′ → p2 ∨. . .∨pn ∨p′1 ∨. . .∨p′n_′

missä kaikki p:t ja q:t ovat faktoja.

”Jos yhden ehdon oikealla puolella on sama kuin toisen vasemmalla, niin ne voidaan pyyhkiä pois ja loput yhdistää uudeksi ehdoksi.”

Muuttujat mukaan resoluutiosääntöön:

• Kummankin loogisen ehdon edessä on

”kaikilla”-kvanttori ∀ ⃗X, ∀ ⃗X′ jokaiselle ehdon muuttujalle, kuten kalvoilla IV.5.2.

• Nimetään loogisten ehtojen muuttujat uudelleen siten, ettei sama nimi esiinny molemmissa: ∀⃗Y , ∀ ⃗Y ′.

• Luetaan sivuehto ”. . . samastuvat yleisimmällä samastimella θ”.

• Johtopäätökseen sovelletaan tätä samastinta θ.

• Johtopäätöksen edessä on kvanttori ∀ ⃗Z jokaiselle siihen jääneelle muuttujalle.

Miten nämä implikaatiot liittyvät Prologiin?

• Kalvoilla IV.5.2 sääntö

p:-q1,. . . ,qm.

tulkittiin loogisesti ∀ ⃗X.q1 ∧ . . . ∧ qm → p. • Koska tyhjä konjunktio (”ja”) on tosi, niin puuttuva vartalo (eli m = 0) eli fakta p. on ∀ ⃗X.tosi → p eli ∀ ⃗X.p.

• Koska tyhjä disjunktio (”tai”) on epätosi, niin puuttuva pää on ∀ ⃗X.q1 ∧ . . . ∧ qm → epätosi eli ∀ ⃗X.¬(q1 ∧ . . . ∧ qm) (missä ’¬’ on negaatio eli ”ei”), eli ¬∃ ⃗X.q1 ∧ . . . ∧ qm (missä ’∃’ on ”on olemassa” -kvanttori). ”Ei ole olemassa sellaisia arvoja muuttujille ⃗X, joilla ongelmat q1, . . . , qm ratkeaisivat yhtä aikaa.” Tämä on maali

?-q1,. . . ,qm.

jota yritetään kumota vastauksella ”onpas; esimerkiksi arvot. . . ”.

Resoluutio onkin refutaatio- eli ristiriitatäydellinen päättelysääntö: jos säännöt ja maali eivät voi olla yhtä aikaa totta, niin sen voi osoittaa resoluutioaskeleilla.

Prologissa resoluutioaskel saa siis muodon

sääntö: p:-q1,...,qm.

maali: ?-q′1,...,q′m_′.

sivuehto: säännön päällä p ja ensimmäisellä alimaalilla q′1 on yleisin samastin θ.

Ensin säännön muuttujat on kuitenkin nimetty erilleen maalin muuttujista.

uusi maali: ?-q1θ,...,qmθ,q′2θ,...,q′m_′θ.

Maali toimii siis (ensimmäistä) ratkaisuaan odottavien alimaalien pinona.

Kun saavutetaan tyhjä maali ?-., on löydetty ristiriita tosi → epätosi. Vastaesimerkkinä kerrotaan todistuksen kuluessa kertynyt sidonta. Kun vielä päätetään

• kokeilla sääntöjä niiden kirjoitusjärjestyksessä

• peruuttaa edelliseen maaliin ja sen seuraavaan sääntöön kun nykyinen maali osoittautuu mahdottomaksi

päästään kalvojen IV.5.1 suorituskoneeseen.

pajakari$ /opt/pl/bin/pl
Welcome to SWI-Prolog (Version 4.0.1)
Copyright (c) 1990-2000 University of Amsterdam.
Copy policy: GPL-2 (see www.gnu.org)

For help, use ?- help(Topic). or ?- apropos(Word).

?- consult(’sum.pl’).
% sum.pl compiled 0.00 sec, 1,112 bytes

Yes
?- trace,p(X,Y,s(s(s(o)))).
   Call: (7) p(_G282, _G283, s(s(s(o)))) ? creep
   Exit: (7) p(o, s(s(s(o))), s(s(s(o)))) ? creep

X = o
Y = s(s(s(o))) ;
   Redo: (7) p(_G282, _G283, s(s(s(o)))) ? creep
   Call: (8) p(_G378, _G283, s(s(o))) ? creep
   Exit: (8) p(o, s(s(o)), s(s(o))) ? creep
   Exit: (7) p(s(o), s(s(o)), s(s(s(o)))) ? creep

X = s(o)
Y = s(s(o)) ;
   Redo: (8) p(_G378, _G283, s(s(o))) ? creep
   Call: (9) p(_G380, _G283, s(o)) ? creep
   Exit: (9) p(o, s(o), s(o)) ? creep
   Exit: (8) p(s(o), s(o), s(s(o))) ? creep
   Exit: (7) p(s(s(o)), s(o), s(s(s(o)))) ? creep

X = s(s(o))
Y = s(o) ;
   Redo: (9) p(_G380, _G283, s(o)) ? creep
   Call: (10) p(_G382, _G283, o) ? creep
   Exit: (10) p(o, o, o) ? creep
   Exit: (9) p(s(o), o, s(o)) ? creep
   Exit: (8) p(s(s(o)), o, s(s(o))) ? creep
   Exit: (7) p(s(s(s(o))), o, s(s(s(o)))) ? creep

X = s(s(s(o)))
Y = o ;
   Redo: (10) p(_G382, _G283, o) ? creep
   Fail: (10) p(_G382, _G283, o) ? creep
   Fail: (9) p(_G380, _G283, s(o)) ? creep
   Fail: (8) p(_G378, _G283, s(s(o))) ? creep
   Fail: (7) p(_G282, _G283, s(s(s(o)))) ? creep

No
[debug] ?-

Edellä todistussemantiikka kuvattiin samassa järjestyksessä kuin kalvojen IV.5.1 suoritussemantiikassa, jossa sääntöjä käytetään ”takaperin” eli niin-osasta jos-osaan.

Kalvojen IV.5.2 mallisemantiikassa sääntöjä käytettiinkin ”etuperin”. Vastaava todistussemantiikka saadaan lukemalla Prolog-lause p(t(i,1), . . . , t(i,k)) :- q(i,1), . . . , q(i,m_i). eli ∀ ⃗X.q(i,1) ∧ . . . ∧ q(i,n) → p(t(i,1), . . . , t(i,k)) (ei enää totuudesta puhuvana väitteenä vaan) todistuvuudesta puhuvana päättelysääntönä q(i,1) . . . q(i,m_1) p(t(i,1), . . . , t(i,k)) joka luetaan ”jos olet saanut todistuksen Ki jokaiselle oletukselle q(i,j), niin voit yhdistää ne todistukseksi K johtopäätökselle p(t(i,1), . . . , t(i,k))”.

Muuttujat tulkitaan kalvojen IV.5.1 hahmojen tavoin:

• Jokainen Ki todistaa jokin muuttujattoman muodon q′i omasta johtopäätöksestään q(i,j).

• Näiden q′i on yhdessä sovittava säännön hahmoon.

• Sovittamisella määräytyy johtopäätöksen todistettu muuttujaton (?) muoto.

Tässä ohjelman luennassa päättelysääntöinä kalvojen IV.5.1 suoritussemantiikka voidaan selittää seuraavasti:

• Todistusta etsitäänkin ”alhaalta ylöspäin” eli johtopäätöksistä oletuksiin.

• Etsinnän kuluessa muuttujia sidotaan samastuksella väin sen verran kuin on välttämätöntä kokeiltavan säännön käyttämiseksi.

• Sääntöjä kokeillaan kirjoitusjärjestyksessä i = 1, . . . , n. • Oletukset yritetään todistaa

kirjoitusjärjestyksessä j = 1, . . . , mi.

Entä mikä on tässä luennassa sellainen looginen päättelysäännöstö, jossa annetun väitteen todistusta etsittäessä ei koskaan ole kuin olennaisesti yksi mahdollisuus?

Sehän on ohjelmointikielen suoritusmekanismin abstrakti määrittely ”vaihtoehdottomana logiikkana”! Esimerkkinä kalvojen II.5 Scheme-ydinkielen määrittely päättelysäännöillä.

Vastaava malliteoria on kielen denotationaalinen eli tarkoitesemantiikka.

IV.6 Negaatio tarkemmin

Kalvojen IV.5 puhtaassa Prologissa ei voi ilmaista negatiivista tietämystä ”. . . ei ole. . . ”.

• Olisi yksinkertaista kirjoittaa faktoja ”henkilö X sairastaa tautia Y ” ja

• määritellä ”henkilö Z on terve jos hän ei sairasta mitään tautia”.

Tämän vuoksi Prologiin on lisätty kalvojen IV.4.4 negaatio epäonnistumisena [B01,§5.3-5.4]: vartalossa voi esiintyä alimaali

not(p)

jonka haluttu luenta on ”väite p ei päde”.

Seuraavaksi tarkastellaan mikä on sen todellinen luenta kalvojen IV.5.1—IV.5.3 näkökulmista.

IV.6.1 Suoritussemantiikassa

Kalvojen IV.5.1 suoritussemantiikassa kutsu not(p) tehdään seuraavasti:

1. Yritetään suorittaa negaation sisällä oleva kutsu p nykyisellä sidonnalla θi.

2. Jos sisäkutsu p epäonnistuu, niin koko kutsu

not(p) onnistuu.

Sidontana pysyy θi.

(Eihän epäonnistunut p voinut kasvattaa sidontaa θi. Myöskään myöhempi peruutus takaisin kutsuun not(p) ei tuota uusia vaihtoehtoja.)

3. Muuten koko kutsu epäonnistui (ja peruutetaan hakemaan vaihtoehtoista θi).

Sairausesimerkkimme on nyt käytännössä:

sairastaa(leopold_duke_of_albany,hemofilia).
sairastaa(rupert,hemofilia).
sairastaa(frederick,hemofilia).
sairastaa(alexis,hemofilia).
sairastaa(waldemar,hemofilia).
sairastaa(henry,hemofilia).
sairastaa(leopold,hemofilia).
sairastaa(maurice,hemofilia).
sairastaa(alfonso,hemofilia).
sairastaa(gonzalo,hemofilia).
sairastaa(luennoija,kevätflunssa).

terve(X):-not(sairastaa(X,_)).

?- trace,terve(luennoija).
   Call: (7) terve(luennoija) ? creep
^ Call: (8) not(sairastaa(luennoija, _G319)) ? creep
   Call: (9) sairastaa(luennoija, _G319) ? creep
   Exit: (9) sairastaa(luennoija, kevätflunssa) ? creep
^ Fail: (8) not(sairastaa(luennoija, _G319)) ? creep
   Fail: (7) terve(luennoija) ? creep

No

?- trace,terve(teräsmies).
   Call: (7) terve(teräsmies) ? creep
^ Call: (8) not(sairastaa(teräsmies, _G313)) ? creep
   Call: (9) sairastaa(teräsmies, _G313) ? creep
   Fail: (9) sairastaa(teräsmies, _G313) ? creep
^ Exit: (8) not(sairastaa(teräsmies, _G313)) ? creep
   Exit: (7) terve(teräsmies) ? creep

Yes

Tähän saakka vastaukset olivat järkeviä, mutta. . .

?- trace,terve(X).
   Call: (8) terve(_G225) ? creep
^ Call: (9) not(sairastaa(_G225, _G284)) ? creep
   Call: (10) sairastaa(_G225, _G284) ? creep
   Exit: (10) sairastaa(leopold_duke_of_albany, hemofilia) ? creep
^ Fail: (9) not(sairastaa(_G225, _G284)) ? creep
   Fail: (8) terve(_G225) ? creep

No

Kalvoilla IV.5.3 Prolog-maali ?- q. luettiin ”onko olemassa sellaisia vapaiden muuttujien arvoja, joilla q pätee (ja jos on niin mitkä ne ovat)?”

Tarkoitimme maalillamme ?- terve(X). siis ”hae kaikki terveet ihmiset X”.

Saimme vastauksen No eli ”heitä ei ole”:

1. Normaali suoritus johti ensin uuteen maaliin

?- not(sairastaa(X, )).

2. Valittu negaation suoritustapa teki siitä maalin

?- sairastaa(X, ).

3. Tämän maalin Yes/No-vastaus käännetään nurin.

Mutta tämä maalihan luetaankin ”sairastaako joku jotakin?”

Siis alkuperäinen maali kysyykin ”ovathan kaikki terveitä?”

Tämän vuoksi negaatioon epäonnistumisena ei voi luottaa jos negatoidussa alimaalissa on kutsuhetkellä vapaita muuttujia.

Maalin implisiittinen ∃X.terve(X) vaihtuu näet negaation ajaksi muotoon ∀X, .¬sairastaa(X, ).

IV.6.2 Mallisemantiikassa

Kalvojen IV.6.1 negatoitujen muuttujien ongelmaa voi yrittää selittää kalvojen IV.5.2 mallisemantiikalla.

Positiivinen muuttujaton maali ?- p. vastaa

Yes/No-kysymykseen ”p ∈ ML”.

Negatiivinen muuttujaton ?- not(q). vastaa kysymykseen ”q ∉ ML”.

Tämän toteuttu negaatio kyllä osaa.

Positiivinen muuttujallinen p on hahmo joka edustaa kaikkia niitä muuttujattomia p′ joilla vastaus on Yes.

Negatiivinen muuttujallinen q edustaa kaikkia q′ joilla vastaus on No.

Tätä toteutettu negaatio ei osaa: se tarkistaa vain onko sellaisia olemassa.

HUOM! Tämä selittää vain käyttäjän kyselyihin kirjoittamat negaatiot. Ohjelmaan kirjoitetut negaatiot saattavat (rekursioon osallistuessaan) jopa tuhota pienimmän mallin ML yksikäsitteisyyden!

IV.6.3 Todistussemantiikassa

Kalvojen IV.5.3 todistussemantiikassa kalvoilla IV.6.1 valittu negaation toteutus on uusi päättelysääntö

Väite not(p) voidaan todistaa osoittamalla ettei väitettä p voi todistaa.

Tämä on ns. suljetun maailman oletus (Closed World Assumption, CWA). Sitä käytetään paljon tekoälyssä.

Koska todistaa sai vain sellaista mikä on varmaa on tämä oletus (täydellisessä päättelyjärjestelmässä)

”koska p ei ole varmaa niin se ei ole totta”.

Ohjelman ”maailma(nkuva)” sulkee siis pois epävarmuuden: jos ohjelmalla ei ole tarpeeksi tietoa väitteestä p, niin se ei ohjelman mielestä voi pitää paikkaansa.

Tietojenkäsittelyllisesti perusteltavissa: ohjelman on reagoitava oman tietämyksensä pohjalta vaikkei se tiedäkään kaikkea ulkomaailmasta.

Tämä on ns. epämonotoninen päättelysääntö: uuden tiedon lisääminen saattaa kumota aiempia päätelmiä.

IV.7 Haun ohjailu

Kalvojen IV.4.4 Prolog-negaatio toteutettiin kalvoilla IV.4.5 ohjailemalla Prolog-hakukoneen toiminta kirjastopredikaateilla !/0 ja fail/0.

Tarkastellaan katkaisupredikaattia !/0 tarkemmin. Olkoon ratkaistavana maali

?-g,....

ja kokeiluvuorossa ohjelman sääntö

p:-q1,...,qm,!,r1,...,rn.

joka vaikuttaa sopivalta katkaisuun saakka, eli on saatu sidonta θm (samastamalla g ja p sidonnaksi θ0, laajentamalla se kutsulla q1 sidonnaksi θ1,. . . ).

• Suoritus etenee normaalisti eteenpäin yli katkaisukohdan vartalon loppuosaan r1,...,rn. sidonnan θm muuttumatta.

• Sivuvaikutuksena etsintäkone hävittää kaikki kutsun g jäljellä olleet vaihtoehtoiset ratkaisutavat:

– vartalon alkuosan q1,...,qm

– loput ohjelmasta.

• Jos suoritus myöhemmin peruuttaa taaksepäin yli katkaisukohdan, niin koko nykyinen maali epäonnistuu heti; sen ensimmäistä kutsua g ei voida enää ratkaista toisin.

Suoritus jatkuu eteenpäin edellisestä maalista, eli siitä jossa g ei vielä ollut ensimmäisenä.

Kalvojen IV.4.5 negaation toteutus on esimerkki säännöstä (uutta) muotoa

jos. . . niin . . . muuten. . .

”Jos P onnistuu niin epäonnistu muuten onnistu.”

Tähän on tarjolla syntaktista sokeria muodossa

P -> fail ; true.

Tässä väliin kirjoitettu funktori ->/2 on jos. . . niin. . . .

Väliin kirjoitettu funktori ;/2 taas on tai.

+ Katkaisu laajentaa Prolog-kielen ilmaisuvoimaa.

+ Katkaisujen lisääminen parantaa Prolog-ohjelman tehokkuutta koska turha peruutus jää pois.

− Katkaisun käyttö tekee Prolog-ohjelmasta (lopullisesti) sellaisen jonka sääntöjä on seurattava kirjoitusjärjestyksessä.

Katkaisu on

vihreä jos se tehostaa ohjelmaa muuttamatta sen merkitystä (= löytämiä vastauksia)

punainen jos ohjelman merkitys muuttuu.

Punaisia katkaisuja tulee käyttää varoen, koska deklaratiiviselta näyttävään ohjelmaan on nyt ilmaantunut proseduraalisia riippuvuuksia!

Kalvojen IV.3 2-3-puuesimerkkiin voidaan lisätä vihreitä katkaisuja koska sen säännöt ovat (vertailujen nojalla) toisensa poissulkevia:

ins23(t(T0,Y,TY,Z,TZ),X,T):-
       X<Y,
       !,
       ins23(T0,X,TX),
       put3l(TX,Y,TY,Z,TZ,T).
ins23(t(T0,Y,TY,Z,TZ),X,T):-
       Y<X,
       X<Z,
       !,
       ins23(TY,X,TX),
       put3m(T0,Y,TX,Z,TZ,T).
ins23(t(T0,Y,TY,Z,TZ),X,T):-
       Z<X,
       ins23(TZ,X,TX),
       put3r(T0,Y,TY,Z,TX,T).

Mutta jos koodia optimoi edelleen poistamalla toisesta säännöstä (nyt tarpeettoman) vertailun Y<X, niin ensimmäisen säännön katkaisusta tuleekin punainen:

• sääntöjä ei voi enää järjestellä vapaasti uudelleen

• ensimmäisen säännön katkaisua ei saa enää poistaa.

Tässä punainen katkaisu ei voittane paljoakaan.

On myös perusteltuja punaisia katkaisuja: samassa esimerkissä säännöt

put2l(c(T0,X,TX),Y,TY,
    t( T0,X,TX, Y,TY)).
put2l(b(T0,X,TX),Y,TY,
    b(b(T0,X,TX),Y,TY)).
put2l(t(T0,X,TX,Y,TY),Z,TZ,
    b(t(T0,X,TX,Y,TY),Z,TZ)).

sanovat ”jos vasen alipuu on kasvanut korkeutta, niin tee asialle jotakin; muuten se käy sellaisenaan”.

Tämä voidaan ilmaista luontevasti, lyhesti ja tehokkaasti

put2l(c(T0,X,TX),Y,TY,
    t( T0,X,TX, Y,TY)):-
       !.
put2l(T,Y,TY,b(T,Y,TY)).

punaisella katkaisulla.

Punaisia katkaisuja käytettäessä on syytä noudattaa erityistä varovaisuutta ja huolellisuutta. . .

Toinen tapa ohjailla hakua on kirjoittaa sellaisia Prolog-ohjelmia, joissa sääntöjä valitaan sen mukaan, millaista tyyppiä eri muuttujien arvot nyt ovat.

Prolog on samaan tapaan piilevästi tyypitetty kieli kuin Scheme (kalvot II.4).

Tyyppien tutkimiseen on seuraavat kirjastopredikaatit [B01,7.1]:

atom(X) kertoo, onko X jakamaton atomi —

Prolog-vastine kalvojen II.4.3 Scheme-symboleille.

Kalvojen IV.4.1 tyhjä Prolog-lista [] on myös atomi.

integer(X) kertoo, onko X kokonaisluku

(kalvot IV.4.2).

float(X) kertoo, onko X liukuluku.

number(X) kertoo, onko X kokonais- tai liukuluku.

atomic(X) kertoo, onko X atomi tai luku.

compound(X) kertoo, onko X rakenteinen termi, eli

muotoa f(. . .).

Myös kalvojen IV.4.1 epätyhjät Prolog-listat ovat rakenteisia. Niissä funktorina f on sama ’.’/2 joka nähtiin jo kalvojen II.8 Scheme-listoissa.

var(X) kertoo, onko X tällä hetkellä sellainen

muuttuja, jota ei vielä ole sidottu.

Jos X on sidottu, mutta vain johonkin toiseen vapaaseen muuttujaan Y , niin tätä yhteyttä ”X ja Y ovat vapaat vapaat mutta samat” ei vielä pidetä sidontana.

Näin var(X) kertoo esimerkiksi, voiko X toimia tulospositiona — jolloin se ei voi olla syötepositio. . .

nonvar(X) kertoo käänteisen ominaisuuden: joko

X ei ole muuttuja tai sitten se on sellainen muuttuja, joka on tällä hetkellä sidottu johonkin ”oikeaan” arvoon, eli atomiin, numeroon tai rakenteiseen termiin.

IV.8 Prolog ja tietokannat

Kalvojen III yksinkertaisessa Prolog-tulkissa säännöt ja faktat talletettiin alkeelliseen ”tietokantaan” – listaan kirjoitusjärjestyksessä.

Myös oikeaan Prolog-toteutukseen kuuluu sisäinen tietokanta jossa sääntöjä säilytetään – edelleen järjestyksessä siltä varalta että ohjelman merkitys riippuisi sääntöjen suoritusjärjestyksestä. . .

Prologilla on myös yhteyksiä ulkoisten relaatiotietokantojen moderneihin kyselykieliin.

Lisäksi useissa Prolog-toteutuksissa on mahdollisuus käyttää jotakin ulkoista relaatiotietokantaa muuttujattomien faktojen tallennukseen. Esimerkiksi SWI Prolog tarjoaa Open DataBase Connectivity (ODBC) -standardin mukaisen kirjaston.

Prolog on siis hyvä korkean tason kieli sellaisiin tiedonhallintatehtäviin, joissa tarvitaan relaatio-operaatioiden lisäksi päättelykykyä.

IV.8.1 Sisäinen tietokanta

[B01,§7.4]

asserta(C) (onnistuu aina kasvattamatta sidontaa

ja) lisää sisäiseen tietokantaan säännön C muiden saman predikaatin sääntöjen eteen.

(Jos C on vartalollinen, niin se on syntaktisista syistä kirjoitettava toisiin sulkuihin).

Peruutettaessa asserta/1 ei anna uusia ”vastauksia”, mutta sen lisäämä C jää tietokantaan.

assertz(C) on kuin asserta/1 paitsi että C

lisätäänkin edellisten perään.

assert(C) on näistä kirjastopredikaateista se jota

suositellaan käytettäväksi kun sisäisen tietokannan järjestyksellä ei ole väliä – esimerkiksi kun lisätään (muuttujaton) fakta eli talletetaan tietoalkioita eikä aliohjelmia.

(Yleensä assert/1 on assertz/1.)

retract(C) poistaa sisäisestä tietokannasta

ensimmäisen säännön (tai faktan) joka sopii hahmoon C ja laajentaa sidontaa siten kuin C sopii poistettuun sääntöön.

Peruutettaessa poistetaan myös toinen, kolmas,. . . kunnes poistettavaa ei enää ole.

Esimerkiksi maali

?- retract((apu(X):-B)).

poistaa predikaatin apu/1 ensimmäisen sellaisen säännön r jolla on vartalo sekä sitoo muuttujan

X säännön r pään parametriin ja muuttujan B

säännön r vartaloon.

Maali

?- retract(apu(Y)),fail.

taas poistaa peruutellen kaikki predikaatin apu/1 vartalottomat faktat.

Tässä on toisto epäonnistuen (failure-driven loop): tietokannalla on tila joka muuttuu sivuvaikutuksena, joten fail-epäonnistumisen aiheuttama peruutus toistaa tätä sivuvaikutusta.

Myös kalvojen IV.4.6 tulostus tapahtui sivuvaikutuksin.

Prolog-standardissa (mutta ei välttämättä kaikissa toteutuksissa)

• consultoitu tai kirjastopredikaatti on staattinen eikä muokattavissa suoritusaikana ilman edeltävää kutsua dynamic/1.

Intuitiivisesti, suorituskelpoinen ohjelmakoodi pidetään sittenkin erillään vapaasti muokattavasta tietokantadatasta, kunnes erikseen julistetaan aikomus muokata jotakin osaa koodista.

• vaikka predikaatin määrittelyä muutettaisiinkin kesken sen suorituksen, jatkuu sen nykyinen suoritus loppuun sillä samalla määrittelyllä jolla alkoikin [DEC96,§10.2].

Intuitiivisesti, tietokantapäivitykset ensin laskevat vanhan tietokannan sisällöstä sen, millainen sisältö uudelle tietokannalle halutaan, ja vasta sitten vaihtavat vanhan sisällön uuteen.

(Sen epäloogisempi mutta usein käytetty vaihtoehto on, että jokainen yksittäinen

assertz/1 ja retract/1 tehdään heti, ja sen tulos

vaikuttaa jo saman päivityksen seuraaviin askeliin.)

IV.8.2 Datalog

Kalvojen IV.8.1 ajatusta datatietokannan pitämisestä erillään säännöistä voi jatkaa edelleen:

• Unohdetaan rakenteiset termit (kuten listat) eli tyydytään pelkkiin atomeihin.

• Ekstensionaalinen tietokanta koostuu pelkästään muuttujattomista – eli vain atomeja sisältävistä – faktoista.

Kukin tällainen predikaatti vastaa tutun relaatiomallin [AHV95,§3] yhtä (muokattavaa) tietokantataulua.

töissä(työntekijä,yritys)
johtaa(johtaja,yritys)
valvoo(konserniyritys,tytäryritys)

• Intensionaalinen tietokanta sisältää ne säännöt, joilla ekstensionaalista kysellään: kyselyt, näkymät,. . .

Säännöissä on eri predikaatit kuin tauluissa.

Päädytään kyselykieleen datalog (database Prolog) [AHV95,§D].

Intensionaaliseen kantaan voidaan kirjoittaa relaatiokyselyjä kuten ”pomot ja alaiset”:

pomo(P,A):-johtaa(P,Y),töissä(A,Y).

         pomo = π1,3σ2=4(johtaa × töissä)

CREATE VIEW pomo AS
      SELECT J.johtaja,T.työntekijä
      FROM johtaa J,
            töissä T,
      WHERE J.yritys=T.yritys

• Sama muuttuja vartalon eri predikaatinkutsuissa vastaa (luonnollista) liitosta (σ...=...(. . . × . . .)).

• Vartalossa esiintyvän muuttujan esiintyminen myös päässä vastaa projektiota (π...).

Jos taas (nimetön) muuttuja esiintyy päässä muttei vartalossa, on sille vaikea keksiä tietokantatulkintaa; siksi sellaisia ei nyt sallita.

Samaa ”relationaalista” ajattelua voi käyttää usein myös Prolog-ohjelmissa: vartalon kukin predikaatinkutsu hakee vastaavasta (ehkä intensionaalisesta) relaatiosta seuraavan monikon, ja koko sääntö kertoo miten nämä vartalon monikot yhdistetään pään mukaisen relaation monikoksi.

Puhtaan relationaalinen tietokanta-ajattelu ei tosin aina ole mahdollista:

• Prologin suoritusjärjestys saattaa vaatia säännöille tai vartalon kutsuille jonkin tietyn järjestyksen, kun taas relaatioajattelussa järjestys on vapaa.

• Vastauksen kokoamisen ajatellaan tapahtuvan datalogissa kokonaisia tietojoukkoja käsittelemällä yksittäisten kutsujen ratkomisen sijaan.

Saman predikaatin eri säännöt vastaavat yhdistettä:

käläkättäjä(X):-töissä(X,nokia).
käläkättäjä(X):-töissä(X,ericsson).

         käläkättäjä =π1σ2=nokiatöissä

∪π1σ2=ericssontöissä

CREATE VIEW käläkättäjä AS
      (SELECT T.työntekijä
       FROM töissä T
       WHERE T.yritys=’nokia’)
      UNION
      (SELECT T.työntekijä
       FROM töissä T
       WHERE T.yritys=’ericsson’)

Kalvoilla IV.4.4 suositeltiin, että Prolog-negaatiota

not(p)

sovellettaisiin vain muuttujattomiin p. Datalogissa riittää, että kaikki negatoidussa p esiintyvät muuttujat esiintyvät samassa vartalossa myös negatoimattomina (ennen itse negaatiota). Silloin negaatio vastaa joukkoerotusta:

työläinen(X):-töissä(X,Y),not(johtaa(X,Y)).
         työläinen = π1(töissä \ johtaa)
CREATE VIEW työläinen AS
      SELECT T.työntekijä
      FROM töissä T
      WHERE T NOT IN johtaa

Näin datalog jossa on negaatio muttei rekursiota on ”relaatiotäydellinen” kyselykieli. (Ilman negaatiota ei saada erotusta.)

Rekursiiviset datalog-säännöt mahdollistavat sellaiset kyselyt, joita ei voi enää kirjoittaa perusrelaatiomallissa, kuten ”ne tytäryritykset B jotka konserni A omistaa (myös epäsuorasti)”:

omistaa(A,B):-valvoo(A,B).
omistaa(A,B):-valvoo(A,C),omistaa(C,B).

Nyt haetaan relaatioiden välisen yhtälön

omistaa = valvoo ∪ π1,4σ2=3(valvoo × omistaa)

ratkaisua omistaa = . . .

Kaikista ratkaisuista halutaan pienin koska

• seurataan kalvojen IV.5.2 ajatusta pienimmästä mahdollisesta mallista joten

• halutaan pienin mahdollinen vastausrelaatio.

Osoittautuu että tämä pienin ratkaisu voidaan laskea kasvattamalla aluksi tyhjää tulosta yhtälön oikealla puolella kunnes se ei enää muutu:

sitten := ∅; repeat

ennen := sitten;

sitten := valvoo ∪ π1,4σ2=3(valvoo × ennen); until ennen = sitten

(Parempia mutta mutkikkaampia algoritmeja on!)

Tämä operaatio on itse asiassa ekstensionaalisen relaation valvoo transitiivinen sulkeuma (transitive closure, TC): ⟨a, b⟩ ∈ omistaa jos ja vain jos on ketju ⟨a, c1⟩ ∈ valvoo, ⟨c1, c2⟩ ∈ valvoo, . . . , ⟨ck, b⟩ ∈ valvoo. Relaatiomallia onkin ehdotettu laajennettavaksi TC-operaatiolla [DD98,§10.6].

SQL3-standardiluonnoksessa tämä laajennus on mukana rekursiivisina näkyminä [DD98,s.395] (syntaksi luennoijan oma arvaus):

CREATE VIEW omistaa AS
      valvoo
      UNION
      (SELECT V.konserniyritys,O.tytäryritys
       FROM valvoo V,
             omistaa O
       WHERE V.tytäryritys=O.konserniyritys)

Negaation sovittaminen tähän rekursioon on yhä ongelma (samoin kuin kalvoilla IV.6).

Patologinen esimerkki on

p:-not(q).
q:-not(p).

jossa negaatio osallistuu rekursioon.

Kalvojen IV.5.2 mukaan ∅ ei vielä ole malli: not(q) on totta mutta p ei.

Hieman suurempi {p} on jo malli: not(p) ei ole totta.

Mutta myös {q} on malli, eikä siis ole vain yhtä pienintä mallia.

Datalog-ajattelussa saadaan yhtälöpari

p = full \q
q = full \p

missä full = {⟨⟩} on relaatiovakio: kaikki 0-paikkaiset monikot.

Jos (yleistetty) algoritmi yrittää laskea sille pienintä ratkaisua (jota ei siis ole), se jää sykkimään:

kierros p q 0 ∅ ∅ 1 full full 2 ∅ ∅ . . .

Myös vastaava Prolog-suoritus jää ikuiseen silmukkaan.

Prolog ja datalog ovat uusien deduktiivisten eli (epätriviaalia) päättelyä suorittavien tietokantojen keskeisiä perusasioita.

IV.8.3 Vastausjoukot

Kalvojen IV.8 tietokanta-ajattelua tuetaan Prolog-ohjelmoinnissa kirjastopredikaateilla, joilla kaikki kyselyn vastaukset voidaan kerätä yhteen listaan ilman että ne pitäisi poimia yksi kerrallaan [B01,§7.6]:

findall(X,P ,L) on intuitiivisesti ”kerää listaksi

L kaikki sellaiset X joilla P ”.

Toisin sanoen, kutsu P ratkaistaan kaikin mahdollisin tavoin, ja joka löydetyllä ratkaisulla θ (johonkin kohtaan) listaan L lisätään Xθ. Sama alkio voi siis toistua.

”Tulostusasu” X voi olla mikä tahansa termi joka sisältää (mm.) kutsun P muuttujia; nämä täytetään vastaavilla löydetyillä arvoilla.

findall(esimies(P),pomo(P,A),L) antaa listan L

kaikista esimiehistä.

L := SELECT X

FROM P

Jos kyselyssä P on muuttujia jotka eivät esiinny asussa X (kuten tässä A) niin niiden eri arvot (tässä alaiset) katoavat – ne projisoitiin pois.

Jos P ei ratkea on L = [].

bagof(X,P ,L) on muuten kuten findall/3, mutta

ne kyselyn P muuttujat Y jotka puuttuvat asusta X käsitellään kuten Prologissa (eikä kadoteta).

Silloin jokaiselle muuttujan Y eri arvolle a, jolla P ratkeaa, kootaan oma epätyhjä lista L vastaavia tuloksia X.

La := SELECT X

FROM P
WHERE Y = a

Nykyinen a löytyy muuttujan Y arvona. Peruutus siirtyy seuraavaan a, tai epäonnistuu, jos niitä ei enää ole.

Notaatiolla bagof(X,Y ^P ,L) voi määrätä muuttujan Y käyttäytymään sittenkin kirjastopredikaatin findall/3 tapaan.

setof(X,P ,L) on muuten kuin bagof/3, mutta L on

järjestetty ja sisältää kunkin X vain kerran.

La := SELECT DISTINCT X

FROM P
WHERE Y = a
ORDER BY X

Kirjastossa on termien vertailuun predikaatti

@</2 (sekä @=<, @>, @>=, == ja \==).

Kirjallisuutta

AS96 H.Abelson&G.J.Sussman: Structure and Interpretation of Computer Programs. Second Edition. MIT Press, 1996.

AHV95 S.Abiteboul,R.Hull,V.Vianu: Foundations of Databases. Addison Wesley, 1995.

AHU83 A.V.Aho,J.E.Hopcroft&J.D.Ullman: Data Structures and Algorithms. Addison Wesley, 1983.

B01 I.Bratko: Prolog Programming for Artificial Intelligence. Third Edition. Addison Wesley, 2001.

CM87 W.F.Clocksin&C.S.Mellish: Programming in Prolog. Third, Revised and Extended Edition. Springer-Verlag, 1987.

CM98 G.Cousineau&M.Mauny: The Functional Approach to Programming. Cambridge University Press, 1998.

DD98 C.J.Date&H.Darwen: Foundation for Object/Relational Databases: The Third Manifesto. Addison Wesley, 1998.

DEC96 P.Deransart,A.Ed-Dbali&L.Cervoni: Prolog:

The Standard. Reference Manual. Springer-Verlag, 1996.

D96 R.K.Dybvig: The Scheme Programming Language: ANSI Scheme. Second Edition. Prentice Hall, 1996.

H92 D.Harel: Algorithmics. Second Edition. Addison Wesley, 1992.

H97 J.R.Hindley: Basic Simple Type Theory.

Cambridge University Press, 1997.

KCR98 R.Kelsey, W.Clinger&J.Rees (toim.): Revised5 Report on the Algorithmic Language Scheme. Higher-Order and Symbolic Computation 11(1), 1998 tai ACM SIGPLAN Notices 33(9), 1998.

L87 J.W.Lloyd: Foundations of Logic

Programming. Second, Extended Edition. Springer-Verlag, 1987.

P02 B.C.Pierce: Type Theory and Programming Languages. MIT Press, 2002.

S98 S.Slade: Object-Oriented Common Lisp.

Prentice Hall, 1998.

SS86 L.Sterling&E.Shapiro: The Art of Prolog. MIT Press, 1986.

S97 B.Stroustrup. The C++ Programming Language. Third Edition. Addison-Wesley, 1997.

T99 S.Thompson: Haskell: The Craft of Functional Programming. Second Edition. Addison-Wesley, 1999.